

# FORMAT

**FORMAT** *[[rep]] parameter...*

## Function

The FORMAT statement is used to specify input and output parameter settings. Settings specified with a FORMAT statement override (at compilation time) default settings in effect for the session that have been set by a GLOBALS command, SET GLOBALS statement, or by the Natural administrator. These settings may in turn be overridden by parameters specified in a DISPLAY, INPUT, PRINT, WRITE, WRITE TITLE, or WRITE TRAILER statement.

The settings remain in effect until the end of a program or until another FORMAT statement is encountered.

A FORMAT statement does not generate any executable code in the Natural program. It is not executed in dependence of the logical flow of a program. It is evaluated during program compilation in order to set parameters for compiling DISPLAY, WRITE, and INPUT statements. The settings defined with a FORMAT statement are applicable to all DISPLAY, WRITE, and INPUT statements which follow. Multiple FORMAT statements are permitted within a program, but only one per report.

## Report Specification - *rep*

The notation (*rep*) may be used to specify the identification of the report for which the FORMAT statement is applicable. A value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified. If (*rep*) is not specified, the FORMAT statement will be applicable to the first report (report 0).

## Parameters

The parameters can be specified in any order and must be separated by one or more spaces. A single entry must not be split between two statement lines.

Field sensitive parameter settings applied here will only be regarded for variable fields used in a INPUT, WRITE, DISPLAY or PRINT statement of the selected report. They do not apply for text-constants used in any of the mentioned statements.

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>      /*      Output
END-DEFINE      /*      Produced
FORMAT PM=I      /*      -----
WRITE 'Text'      VARI      /*      Text 4321
WRITE 'Text' (PM=I) VARI      /*      txeT 4321
END
  
```

See the section Session Parameters in the Natural Reference documentation for a description of the parameters which may be used.

## Example

```

/* EXAMPLE 'FMTEX1:' FORMAT
/*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 POST-CODE
  2 COUNTRY
END-DEFINE
/*****
FORMAT AL=7      /* ALPHANUMERIC FIELD OUTPUT LENGTH
      FC=+      /* FILLER CHARACTER FOR FIELD HEADER
      GC=*      /* FILLER CHARACTER FOR GROUP HEADER
      HC=L      /* HEADER LEFT JUSTIFIED
      IC=<<      /* INSERT CHARACTERS
      IS=ON      /* IDENTICAL SUPPRESS ON
      TC=>>      /* TRAILING CHARACTERS
      UC==      /* UNDERLINE CHARACTER
      ZP=OFF     /* ZERO PRINT OFF
/*****
LIMIT 5
READ EMPLOY-VIEW BY NAME
  DISPLAY NOTITLE NAME 3X CITY 3X POST-CODE 3X COUNTRY
END-READ
/*****
END

```

NAME++++++	CITY++++++	POSTAL+++++ ADDRESS	COUNTRY++++
=====	=====	=====	=====
<<ABELLAN>>	<<MADRID >>	<<28014 >>	<<E >>
<<ACHIESO>>	<<DERBY >>	<<DE3 4TR>>	<<UK >>
<<ADAM >>	<<JOIGNY >>	<<89300 >>	<<F >>
<<ADKINSO>>	<<BEDFORD>>	<<1730 >>	<<USA>>
	<<FRAMING>>	<<1701 >>	